

```

*****
/*      MACRO: FILESTMP.WCM
/*      PURPOSE: Inserts the document's filename in the position and format specified.
*****
Application (A1; "WordPerfect"; Default; "EN")

LeftJust := 1
NameOnly := 1
FtrA := 1
ChangeFont := 0

// Do this only at the main document screen
While (?Substructure <>0)
    SubstructureExit ()
EndWhile

DialogDefine ("Dlg1"; 50; 50; 238; 120; 16; "File Stamp Options")
DialogAddGroupBox ("Dlg1"; "Gb2"; 8; 5; 92; 45; "&Stamp")
DialogAddRadioButton ("Dlg1"; "Rb4"; 18; 18; 72; 11; "Filename &only"; NameOnly)
DialogAddRadioButton ("Dlg1"; "Rb5"; 18; 30; 72; 11; "Filename and &path"; NamePath)
DialogAddGroupBox ("Dlg1"; "Gb1"; 8; 55; 92; 57; "&Justification")
DialogAddRadioButton ("Dlg1"; "Rb1"; 18; 68; 50; 11; "&Left"; LeftJust)
DialogAddRadioButton ("Dlg1"; "Rb2"; 18; 80; 50; 11; "&Centered"; CenterJust)
DialogAddRadioButton ("Dlg1"; "Rb3"; 18; 92; 50; 11; "&Right"; RightJust)
DialogAddGroupBox ("Dlg1"; "Gb3"; 110; 5; 62; 69; "Place&ment")
DialogAddRadioButton ("Dlg1"; "Rb6"; 120; 18; 42; 11; "Header &A"; HdrA)
DialogAddRadioButton ("Dlg1"; "Rb7"; 120; 30; 42; 11; "Header &B"; HdrB)
DialogAddRadioButton ("Dlg1"; "Rb8"; 120; 42; 42; 11; "&Footer A"; FtrA)
DialogAddRadioButton ("Dlg1"; "Rb9"; 120; 54; 42; 11; "Foo&ter B"; FtrB)
DialogAddCheckBox ("Dlg1"; "Cb1"; 110; 82; 55; 11; "C&hange font"; ChangeFont)
DialogAddPushButton ("Dlg1"; 1; 180; 8; 50; 14; OKBtn!; "OK")
DialogAddPushButton ("Dlg1"; 2; 180; 26; 50; 14; CancelBtn!; "Cancel")
Display (On!)
DialogShow ("Dlg1"; "Rb4")
DialogResult := MacroDialogResult
DialogDestroy ("Dlg1")
Display (Off!)

If (DialogResult <>1)
    Go (End@)
EndIf

AlreadyExists := False

If (HdrA = 1)
    StructureName := "Header A"

```

```

    If (?HeaderAOccur <>0)
        Call (StructureExists@)
        HeaderA (Edit!)
    Else
        HeaderA (Create!)
    EndIf
Else
    If (HdrB = 1)
        StructureName := "Header B"
        If (?HeaderBOccur <>0)
            Call (StructureExists@)
            HeaderB (Edit!)
        Else
            HeaderB (Create!)
        EndIf
    Else
        If (FtrA = 1)
            StructureName := "Footer A"
            If (?FooterAOccur <>0)
                Call (StructureExists@)
                FooterA (Edit!)
            Else
                FooterA (Create!)
            EndIf
        Else
            If (FtrB = 1)
                StructureName := "Footer B"
                If (?FooterBOccur <>0)
                    Call (StructureExists@)
                    FooterB (Edit!)
                Else
                    FooterB (Create!)
                EndIf
            EndIf
        EndIf
    EndIf
EndIf
If (AlreadyExists)
    PosDocBottom ()
    HardReturn ()
EndIf
If (ChangeFont = 1)
    FontDlg ()
EndIf
If (LeftJust = 1)
    JustifyLeft ()

```

```

Else
    If (CenterJust = 1)
        JustifyCenter ()
    Else
        If (RightJust = 1)
            JustifyRight ()
        EndIf
    EndIf
EndIf
If (NameOnly = 1)
    InsertFilename ()
Else
    InsertFilenameWithPath ()
EndIf
SubStructureExit ()

If (?Name = "")
    MessageBox (; "File Stamp Complete"; "The filename code has been placed in ^0. You
    must save the file for the filename to be visible."; IconInformation! | HasParameters!;
    StructureName)
EndIf

Label (End@)
Quit

```

```

//*****
//*   ROUTINE: StructureExists@
//*   INPUT VARIABLES: StructureName
//*   OUTPUT VARIABLES: AlreadyExists (boolean)
//*   DESCRIPTION: Displays a message box.
//*****
Label (StructureExists@)
AlreadyExists := True
MessageBox (MsgBoxResult; StructureName + " Already Exists"; "^0 already exists. If you
continue, the filename will be placed at the bottom of ^0."; IconExclamation! | OKCancel! |
HasParameters!; StructureName)
If (MsgBoxResult <> 1)
    Go (End@)
EndIf
Return
//*****

```